

PROJECT 1

Spartan6 - DSP48A1

By: Ziad Tamer Ibrahim

Rtl code

```
module reg_mux #(
    parameter width = 18,
    parameter rsttype = "SYNC",
    parameter regon =1
)(
    input clken,
    input clk,
    input rst,
    input [width-1:0] A,
    output reg [width-1:0] out
);
    generate
        if (regon) begin
            if (rsttype == "SYNC") begin
                always @(posedge clk) begin
                        if (rst) out <= 0;
                        else if (clken)  out <= A;
                end
            end
            else if (rsttype == "ASYNC") begin
                always @(posedge clk or posedge rst) begin
                    if (rst) out <= 0;
                    else if (clken)  out <= A;
                end
            end
        end
        else begin
            always @(*) begin
                out = A;
            end
        end
    endgenerate
endmodule

module add_sub #(
    parameter width = 18
)(
    input [width-1:0] in1,
    input [width-1:0] in2,
    input cin,
    input opmode, // 0 for addition, 1 for subtraction
```

```
    output reg [width-1:0] out,
    output reg cout
);
    always @(*) begin
        if (opmode == 0) begin
            {cout,out} = in1 + in2 + cin; // Addition
        end
        else begin
            {cout,out} = in1 - in2 - cin; // Subtraction
        end
    end
endmodule

module multiplier #(
    parameter width = 18
)(
    input [width-1:0] in1,
    input [width-1:0] in2,
    output  [2*width-1:0] out
);
    assign out = in1 * in2;
endmodule

module DSP #(
    parameter A0REG = 0,
    parameter A1REG = 1,
    parameter B0REG = 0,
    parameter B1REG = 1,
    parameter CREG = 1,
    parameter DREG = 1,
    parameter MREG = 1,
    parameter PREG = 1,
    parameter CARRYINREG = 1,
    parameter CARRYOUTREG = 1,
    parameter OPMODEREG =1,
    parameter CARRYINSEL = "OPMODE5",
    parameter B_INPUT = "DIRECT",
    parameter RSTTYPE = "SYNC"
)(
    //DATA PORTS
    input [17:0] A,
    input [17:0] B,
    input [47:0] C,
    input [17:0] D,
    input CARRYIN,
```

```
    output [35:0] M,
    output [47:0] P,
    output CARRYOUT,
    output CARRYOUTF,
    // CONTROL  INPUT PORTS
    input clk,
    input [7:0] OPMODE,
    // CLOCK ENABLE INPUT PORTS
    input CEA,
    input CEB,
    input CEC,
    input CECARRYIN,
    input CED,
    input CEM,
    input CEOPMODE,
    input CEP,
    // RESET INPUT PORTS
    input RSTA,
    input RSTB,
    input RSTC,
    input RSTD,
    input RSTCARRYIN,
    input RSTM,
    input RSTOPMODE,
    input RSTP,
    // CASCADE PORTS
    input [17:0] BCIN,
    output [17:0] BCOUT,
    input [47:0] PCIN,
    output [47:0] PCOUT
);
wire [17:0] B0_in, A0_reg, B0_reg, D_reg, A1_reg,add1_out, add2_out,B1_in,B1_reg;
wire [47:0] C_reg, concat, P_in , P_reg ;
reg [47:0] X, Z;
wire [35:0] M_in, M_reg;
wire [7:0] OPMODE_reg;
wire carry_in,carry_reg, carryout_in, carryout_reg,cout_dummy;

assign B0_in = (B_INPUT == "DIRECT")? B :(B_INPUT == "CASCADE")? BCIN : 0;

reg_mux #(
    .width(18),
    .rsttype(RSTTYPE),
    .regon(A0REG)
) A0_REG (
```

```
    .clken(CEA),
    .clk(clk),
    .rst(RSTA),
    .A(A),
    .out(A0_reg)
);
reg_mux #(
    .width(18),
    .rsttype(RSTTYPE),
    .regon(A1REG)
) A1_REG (
    .clken(CEA),
    .clk(clk),
    .rst(RSTA),
    .A(A0_reg),
    .out(A1_reg)
);
reg_mux #(
    .width(18),
    .rsttype(RSTTYPE),
    .regon(B0REG)
) B0_REG (
    .clken(CEB),
    .clk(clk),
    .rst(RSTB),
    .A(B0_in),
    .out(B0_reg)
);
reg_mux #(
    .width(18),
    .rsttype(RSTTYPE),
    .regon(B1REG)
) D_REG (
    .clken(CED),
    .clk(clk),
    .rst(RSTD),
    .A(D),
    .out(D_reg)
);
reg_mux #(
    .width(48),
    .rsttype(RSTTYPE),
    .regon(CREG)
) C_REG (
    .clken(CEC),
```

```
    .clk(clk),
    .rst(RSTC),
    .A(C),
    .out(C_reg)
);
reg_mux #(
    .width(8),
    .rsttype(RSTTYPE),
    .regon(OPMODEREG)
) OPMODE_REG (
    .clken(CEOPMODE),
    .clk(clk),
    .rst(RSTOPMODE),
    .A(OPMODE),
    .out(OPMODE_reg)
);
add_sub #(
    .width(18)
) pre_add_sub (
    .in1(D_reg),
    .in2(B0_reg),
    .cin(1'b0),
    .opmode(OPMODE_reg[6]),
    .out(add1_out),
    .cout(cout_dummy)
);

assign B1_in = (OPMODE_reg[4]) ? add1_out : B0_reg;

reg_mux #(
    .width(18),
    .rsttype(RSTTYPE),
    .regon(B1REG)
) B1_REG (
    .clken(CEB),
    .clk(clk),
    .rst(RSTB),
    .A(B1_in),
    .out(B1_reg)
);
multiplier #(
    .width(18)
) multiplier_inst (
    .in1(A1_reg),
    .in2(B1_reg),
```

```
    .out(M_in)
);
reg_mux #(
    .width(36),
    .rsttype(RSTTYPE),
    .regon(MREG)
) M_REG (
    .clken(CEM),
    .clk(clk),
    .rst(RSTM),
    .A(M_in),
    .out(M_reg)
);

assign M = M_reg;
assign BCOUT = B1_reg;
assign concat = {D_reg[11:0], A1_reg[17:0], B1_reg[17:0]};
assign carry_in =   (CARRYINSEL == "CARRYIN")? CARRYIN :
                (CARRYINSEL == "OPMODE5")? OPMODE_reg[5] : 0;

reg_mux #(
    .width(1),
    .rsttype(RSTTYPE),
    .regon(CARRYINREG)
) CYI (
    .clken(CECARRYIN),
    .clk(clk),
    .rst(RSTCARRYIN),
    .A(carry_in),
    .out(carry_reg)
);
always @(*) begin
    case (OPMODE_reg[1:0])
        2'b00: X = 0; // Default case
        2'b01: X = {12'b0, M_reg}; // PCIN
        2'b10: X = PCOUT; // Zero extension of M
        default: X = concat; // Concatenation
    endcase
    case (OPMODE_reg[3:2])
        2'b00: Z = 0; // Default case
        2'b01: Z = PCIN; // PCIN
        2'b10: Z = PCOUT; // PCOUT
        default: Z = C_reg; // C
    endcase
end
```

```
add_sub #(
    .width(48)
) post_add_sub (
    .in1(Z),
    .in2(X),
    .cin(carry_reg),
    .opmode(OPMODE_reg[7]),
    .out(P_in),
    .cout(carryout_in)
);
reg_mux #(
    .width(1),
    .rsttype(RSTTYPE),
    .regon(CARRYOUTREG)
) CYO (
    .clken(CEP),
    .clk(clk),
    .rst(RSTCARRYIN),
    .A(carryout_in),
    .out(carryout_reg)
);
assign CARRYOUT = carryout_reg;
assign CARRYOUTF = carryout_reg;

reg_mux #(
    .width(48),
    .rsttype(RSTTYPE),
    .regon(PREG)
) P_REG (
    .clken(CEP),
    .clk(clk),
    .rst(RSTP),
    .A(P_in),
    .out(P_reg)
);
assign P = P_reg;
assign PCOUT = P_reg;
endmodule
```

Testbench code

```
`timescale 1ns/1ps

module DSP_tb();

  reg clk;
  reg [17:0] A, B, D;
  reg [47:0] C;
  reg CARRYIN;
  reg [7:0] OPMODE;

  reg CEA, CEB, CEC, CECARRYIN, CED, CEM, CEOPMODE, CEP;
  reg RSTA, RSTB, RSTC, RSTD, RSTCARRYIN, RSTM, RSTOPMODE, RSTP;
  reg [17:0] BCIN;
  reg [47:0] PCIN;

  reg [47:0] last_P;
  reg last_CARRYOUT;

  wire [35:0] M;
  wire [47:0] P;
  wire CARRYOUT, CARRYOUTF;
  wire [17:0] BCOUT;
  wire [47:0] PCOUT;

  DSP DUT (
    .A(A), .B(B), .C(C), .D(D), .CARRYIN(CARRYIN),
    .M(M), .P(P), .CARRYOUT(CARRYOUT), .CARRYOUTF(CARRYOUTF),
    .clk(clk), .OPMODE(OPMODE),
    .CEA(CEA), .CEB(CEB), .CEC(CEC), .CECARRYIN(CECARRYIN),
    .CED(CED), .CEM(CEM), .CEOPMODE(CEOPMODE), .CEP(CEP),
    .RSTA(RSTA), .RSTB(RSTB), .RSTC(RSTC), .RSTD(RSTD),
    .RSTCARRYIN(RSTCARRYIN), .RSTM(RSTM), .RSTOPMODE(RSTOPMODE), .RSTP(RSTP),
    .BCIN(BCIN), .BCOUT(BCOUT), .PCIN(PCIN), .PCOUT(PCOUT)
  );

  initial clk = 0;
  always #5 clk = ~clk;

  initial begin
```

```
    $display("=== DSP Reset Verification Test ===");

    A = $random;
    B = $random;
    D = $random;
    C = $random;
    CARRYIN = $random;
    OPMODE = $random;
    BCIN = $random;
    PCIN = $random;

    RSTA = 1;
    RSTB = 1;
    RSTC = 1;
    RSTD = 1;
    RSTCARRYIN = 1;
    RSTM = 1;
    RSTOPMODE = 1;
    RSTP = 1;

    CEA = 0;
    CEB = 0;
    CEC = 0;
    CECARRYIN = 0;
    CED = 0;
    CEM = 0;
    CEOPMODE = 0;
    CEP = 0;


    @(negedge clk);

    if (M !== 0 || P !== 0 || CARRYOUT !== 0 || CARRYOUTF !== 0 || BCOUT !== 0 ||
PCOUT !== 0) begin
        $display(" Reset Test FAILED at time %0t", $time);
        $display("M = %h, P = %h, CARRYOUT = %b, BCOUT = %h, PCOUT = %h", M, P,
CARRYOUT, BCOUT, PCOUT);
        $stop;
    end else begin
        $display(" Reset Test PASSED");
    end

    RSTA = 0; RSTB = 0; RSTC = 0; RSTD = 0;
    RSTCARRYIN = 0; RSTM = 0; RSTOPMODE = 0; RSTP = 0;
```

```
    CEA = 1; CEB = 1; CEC = 1; CECARRYIN = 1;
    CED = 1; CEM = 1; CEOPMODE = 1; CEP = 1;

    $display("=== DSP PATH1 Verification Test ===");

    A = 20;
    B = 10;
    C = 48'd350;
    D = 25;
    OPMODE = 8'b11011101;
    BCIN = $random;
    PCIN = $random;
    CARRYIN = $random;

    repeat (4) @(negedge clk);


    if (BCOUT != 18'h00F || M != 36'h12C || P != 48'h32 || PCOUT != 48'h32 ||
CARRYOUT != 0 || CARRYOUTF != 0) begin
        $display(" DSP Path 1 Test FAILED");
        $display("BCOUT     = %h (expected 000F)", BCOUT);
        $display("M         = %h (expected 012C)", M);
        $display("P         = %h (expected 0032)", P);
        $display("PCOUT     = %h (expected 0032)", PCOUT);
        $display("CARRYOUT  = %b (expected 0)", CARRYOUT);
        $display("CARRYOUTF = %b (expected 0)", CARRYOUTF);

        $stop;
    end else begin
        $display(" DSP Path 1 Test PASSED");
    end

    $display("=== DSP PATH2 Verification Test ===");
    A = 20;
    B = 10;
    C = 48'd350;
    D = 25;
    OPMODE = 8'b00010000;
    BCIN = $random;
    PCIN = $random;
    CARRYIN = $random;

    repeat (3) @(negedge clk);
```

```
    if (BCOUT != 18'h23 || M != 36'h2bc || P != 0 || PCOUT != 0 || CARRYOUT != 0
|| CARRYOUTF != 0) begin
        $display(" DSP Path 2 Test FAILED");
        $display("BCOUT     = %h (expected 0023)", BCOUT);
        $display("M         = %h (expected 02bc)", M);
        $display("P         = %h (expected 0000)", P);
        $display("PCOUT     = %h (expected 0000)", PCOUT);
        $display("CARRYOUT  = %b (expected 0)", CARRYOUT);
        $display("CARRYOUTF = %b (expected 0)", CARRYOUTF);

        $stop;
    end else begin
        $display(" DSP Path 2 Test PASSED");
    end

    $display("=== DSP PATH3 Verification Test ===");

    last_CARRYOUT = CARRYOUTF;
    last_P = PCOUT;

    A = 20;
    B = 10;
    C = 48'd350;
    D = 25;
    OPMODE = 8'b00001010;
    BCIN = $random;
    PCIN = $random;
    CARRYIN = $random;

    repeat (3) @(negedge clk);

    if (BCOUT != 18'ha || M != 36'hc8 || P != last_P || CARRYOUT !=
last_CARRYOUT) begin
        $display(" DSP Path 3 Test FAILED");
        $display("BCOUT     = %h (expected 000a)", BCOUT);
        $display("M         = %h (expected 00c8)", M);
        $display("P         = %h (expected %h)", P, last_P);
        $display("CARRYOUT  = %b (expected 0)", CARRYOUT);

        $stop;
    end else begin
        $display(" DSP Path 3 Test PASSED");
    end

    $display("=== DSP PATH4 Verification Test ===");
```

```
    A = 5;
    B = 6;
    C = 48'd350;
    D = 25;
    PCIN = 48'd3000;
    OPMODE = 8'b10100111;
    BCIN = $random;
    CARRYIN = $random;

    repeat (3) @(negedge clk);

    if (BCOUT != 18'h6 || M != 36'h1e || P != 48'hfe6fffec0bb1 || PCOUT !=
48'hfe6fffec0bb1 || CARRYOUT != 1 || CARRYOUTF != 1) begin
        $display(" DSP Path 4 Test FAILED");
        $display("BCOUT     = %h (expected 000a)", BCOUT);
        $display("M         = %h (expected 00c8)", M);
        $display("P         = %h (expected fe6fffec0bb1)", P );
        $display("PCOUT     = %h (expected fe6fffec0bb1)", PCOUT);
        $display("CARRYOUT  = %b (expected 1)", CARRYOUT);
        $display("CARRYOUTF = %b (expected 1)", CARRYOUTF);

        $stop;
    end else begin
        $display(" DSP Path 4 Test PASSED");
    end

    $stop;
  end
endmodule
```

Do File

vlib work

vlog DSP.v DSP_tb.v

vsim -voptargs=+acc work.DSP_tb

add wave *

run -all

Questasim snippets

Reset Test



Path 1 Test



Path 2 Test



Path 3 Test



Path 4 Test



Transcript

```
VSIM(paused)> do run_dsp.do
# ** Warning: (vlib-34) Library already exists at "work".
# Errors: 0, Warnings: 1
# QuestaSim-64 vlog 2021.1 Compiler 2021.01 Jan 19 2021
# Start time: 18:58:35 on Jul 29,2025
# vlog -reportprogress 300 DSP.v DSP_tb.v
# -- Compiling module DSP
# -- Compiling module DSP_tb
#
# Top level modules:
#       DSP_tb
# End time: 18:58:35 on Jul 29,2025, Elapsed time: 0:00:00
# Errors: 0, Warnings: 0
# vsim -voptargs="+acc" work.DSP_tb
# Start time: 18:58:35 on Jul 29,2025
# ** Note: (vsim-8009) Loading existing optimized design _opt1
# Loading work.DSP_tb(fast)
# Loading work.DSP(fast)
# Loading work.reg_mux(fast)
# Loading work.reg_mux(fast__1)
# Loading work.reg_mux(fast__2)
# Loading work.reg_mux(fast__3)
# Loading work.add_sub(fast)
# Loading work.multiplier(fast)
# Loading work.reg_mux(fast__4)
# Loading work.reg_mux(fast__5)
# Loading work.add_sub(fast__1)
# === DSP Reset Verification Test ===
#  Reset Test PASSED
# === DSP PATH1 Verification Test ===
#  DSP Path 1 Test PASSED
# === DSP PATH2 Verification Test ===
#  DSP Path 2 Test PASSED
# === DSP PATH3 Verification Test ===
#  DSP Path 3 Test PASSED
# === DSP PATH4 Verification Test ===
#  DSP Path 4 Test PASSED
# ** Note: $stop    : DSP_tb.v(194)
#    Time: 140 ns  Iteration: 1  Instance: /DSP_tb
# Break in Module DSP_tb at DSP_tb.v line 194

VSIM(paused)>
```

Constraint file

```
## Clock signal
set_property -dict {PACKAGE_PIN W5 IOSTANDARD LVCMOS33} [get_ports clk]
create_clock -period 10.000 -name sys_clk_pin -waveform {0.000 5.000} -add
[get_ports clk]
```

Elaboration

Schematic



Messages



Synthesis

Schematic



Messages tab



Utilization report



Timing report



Implementation

Device schematic



Messages tab

Timing report



Lint


Questa Lint 2021.1 (...ments/Digital_Design_Diploma/pr1/lint.db)
File Edit View Lint Checks Window Help
Show Collated View
Design
Search: Type ...
Instance
DSP (14)
D:/Documents/Digital_Design_Diploma/pr1/DSP.v [DSP]
module DSP #(
parameter A0REG = 0,
parameter A1REG = 1,
parameter B0REG = 0,
parameter B1REG = 1,
parameter CREG = 1,
parameter DREG = 1,
parameter MREG = 1,
parameter PREG = 1,
parameter CARRYINREG = 1,
parameter CARRYOUTREG = 1,
parameter OPMODEREG =1,
parameter CARRYINSEL = "OPMODE5",
OPMODE5
Schematic 1
Flow Na...
D...
Lint Summary
Schematic 1
Lint Checks
Filter: Type here
Waived
Fixed
Pending
Uninspected
Bug
Verified
Total : 3
Selected : 1
Severity
Status
Check
Alias
Message
Module
Category
State
Owner
STARC Reference
condition_const
Condition expression is a constant. Module DSP, File ...
DSP
Rtl Design Style
open
unassig...
condition_const
Condition expression is a constant. Module DSP, File ...
DSP
Rtl Design Style
open
unassig...
parameter_name_duplicate
Same parameter name is used in more than one mod...
add_sub
Nomenclature...
open
unassig...
1.1.4.3, 3.2.2.2
Transcript
Message Viewer
Lint Checks
Design Metrics
Design Information
Status History
Lint Dashboard